




Linked Data Workshop

Zürich, 22.11.2023

J. Riegelning, K. Rosin, R. Schenker, H. Stalder, D. Truttmann

Inhalte

1. Begrüssung (5')



2. Theorie (15')



3. Praxis (40')



4. Pause (10')




5. Praxis (40')



6. Abschluss (10')



1 Begrüßung



2 Theorie



2.1 Konzepte

Titel? Inhalte? Login?

The screenshot shows the homepage of the Tages-Anzeiger newspaper. At the top left is the logo 'Tages-Anzeiger'. To the right, there is a blue button for '1. Monat für 0 CHF', a 'Login' link, and a 'Menü' icon. Below this is a horizontal navigation bar with categories: Zürich, Schweiz, Ukraine-Krieg, International, Wirtschaft, Sport, Züritipp(s), Das Magazin, Meinungen, Blogs, Kultur, Wissen, and Panorama. The main content area features a large infographic on the left and three news articles on the right.

Infographic: Kinderlose

Category	Percentage
Kinderlose	21,5%
Kinderlose	34,3%

Article 1: Auswertung zum Mietmarkt
Mietende bleiben in zu grossen Wohnungen, weil eine kleinere teurer wäre
Neue Zahlen der Zürcher Kantonalbank zeigen, dass viele kleine Wohnungen über- und grosse Wohnungen unterbelegt sind. Dies, weil sich die Mietenden keinen Umzug leisten können.

Article 2: Nationaltrainer in der Kritik
Doch die Frage bleibt: War es das für Murat Yakin?

Article 3: Analyse zum Zustand des Freisinns
Die FDP sitzt in der Falle – und macht selber alles noch schlimmer
Auch wenn die Grünen am meisten Wähler verloren haben: Das grösste Problem hat die FDP. Was kann diese Partei noch retten? Eine Schadenbilanz in fünf Punkten.
Markus Häfliger

Podcast: «Apropos» – der tägliche Podcast
Corona ist zurück – das muss man jetzt wissen
Wer zurzeit krank ist, hat sich wahrscheinlich mit dem Coronavirus angesteckt. Was es zu beachten gibt – und was die nächsten Wochen auf uns zukommt.

Titel? Inhalte? Login?

21 nov 2023 | Actualizado 08:08 CET | Selección: ESPAÑA

SUSCRIBETE INICIAR SESIÓN

EL PAÍS

EL PERIÓDICO GLOBAL

INTERNACIONAL OPINIÓN ESPAÑA ECONOMÍA SOCIEDAD EDUCACIÓN MEDIO AMBIENTE CIENCIA SALUD TECNOLOGÍA CULTURA BABELIA DEPORTES TELEVISIÓN GENTE EL PAÍS SEMANAL

El nuevo Gobierno DIRECTO

 Pedro Sánchez Presidente del Gobierno	 Nadia Calviño Vicepresidenta primera y ministra de Asuntos Económicos, Comercio y Empresa	 Yolanda Díaz Vicepresidenta segunda y ministra de Trabajo y Economía Social	 Teresa Ribera Vicepresidenta tercera y ministra de Transición Ecológica	 María Jesús Montero Vicepresidenta cuarta y ministra de Hacienda	 José Manuel Albares Ministro de Asuntos Exteriores, UE y Cooperación	 Félix Bolaños Ministro de Justicia y Presidencia
--	--	---	--	---	---	---

←

Sánchez da más poder a su núcleo duro y se rodea de políticos para afrontar una legislatura tensa

CARLOS E. CLUÉ / JOSÉ MARCOS | Madrid

El Ejecutivo, con cargos de confianza y pocas sorpresas, afrontará la tramitación de la amnistía, el debate presupuestario y el modelo territorial

— Todos los ministros del Ejecutivo



Javier Milei, durante la celebración de la victoria electoral al domingo. JUAN IGNACIO RONCORONI (EFE)

Opinión

EDITORIAL

Un Gobierno para el vendaval

Pedro Sánchez refuerza el perfil político de su nuevo Gabinete pero difumina el área económica

La Argentina, otro país

El país ha terminado de demostrar que es, ahora, un país desesperado, porque hay que estar desesperado para votar a

Titel? Inhalte? Login?

The screenshot shows the Asahi News website homepage. At the top, the logo '朝日新聞 DIGITAL' is on the left, and navigation links for 'パレスチナ情勢', 'ウクライナ', '速報', '朝刊', '夕刊', '連載', 'ランキング', and a search icon are on the right. Below the logo, a horizontal menu lists categories: 'トップ', '社会', '経済', '政治', '国際', 'スポーツ', 'オピニオン', 'IT・科学', '文化・芸能', 'ライフ', '教育・子育て', '医療・健康', and '地域'. The main content area features several news items:

- ガザの未来「米とイスラエルで決めるな」 パレスチナ副首相単独会見** (17:00) with a thumbnail of a building.
- 防カメに「かかと落とし」の障害者施設元職員、別の入所者に暴行容疑** (16:30) with a thumbnail of a building.
- 副業のベビーシッター先で姉妹にわいせつ行為の疑い 37歳男を逮捕** (17:00) with a thumbnail of a stone monument.
- スカウト集団「ナチュラル」幹部逮捕 無許可紹介か、収益は暴力団に** (14:29) with a thumbnail of a group of people.
- 温泉ホテル近くを歩くヒグマ 知床で目撃件数が過去最高になったワケ** (16:00) with a thumbnail of a bear.
- 90歳の現役声優が語る、老いない極意 いつまでも必要とされるには** (16:30) with a thumbnail of a person.
- 「レンタル彼氏」が感じた違和感 大学院生は「男性像」を語り始めた** (14:00) with a thumbnail of a person.

On the right side, there is a '速報ニュース' section with a '最新コメント' tab, showing a list of recent comments with timestamps like '1分前' and '2分前'. At the bottom right, there is a 'PR 注目情報' section with a weather forecast for '11月21日(火) の天気' and a '地域を設定' button.


Semantik für Maschinen

- Semantik/Bedeutung von Texten ist für Maschinen schwer erfassbar
- Kontext fehlt oftmals
- Beispiele: Bank, Erde, Jaguar, usw.




Idee: Beziehungen beschreiben

Beziehungen oder Eigenschaften mit sogenannten Triples beschreiben




Idee: Beziehungen beschreiben



Idee: Beziehungen beschreiben

- <rolf>: Im einen Triple Subjekt, im anderen Triple Objekt
- Mehrere Triples ergeben einen Graphen



2.2 Begriffe

Begriffe

- Uniform Resource Identifier (URI)
 - **Eindeutige Identifizierung**: Reales oder Abstraktes
 - Webseiten, Personen, Produkte, Standorte, Eigenschaften, Beziehungen
 - Analog zu Web: **URL** (Uniform Resource Locator)
 - Analog zu Literatur: **ISBN** (International Standard Book Number)

- SPARQL
 - **Abfragesprache für Linked Data**
 - [SPARQL](#) = SPARQL Protocol And RDF Query Language
 - Deutsch: SPARQL Protokoll- und RDF-Abfragesprache

2.3 SPARQL

SELECT, WHERE



- Variable: mit Fragezeichen
- WHERE: Triples
- SELECT
 - Welche Variablen ausgegeben?
 - Stern (*) bedeutet alle Variablen ausgeben
 - SELECT DISTINCT: unterschiedliche Records ausgeben

```
SELECT ?frucht
WHERE
{
  ?frucht <hatFarbe> "gelb" .
  ?frucht <schmeckt> "bitter" .
}
```

```
SELECT DISTINCT *
WHERE
{
  ?frucht <hatFarbe> "gelb" .
  ?gemuese < hatFarbe > "gruen" .
}
```


Satzzeichen



- Nach Triple: **Punkt**
- Subjekt gleich: **Strichpunkt**
- Subjekt und Prädikat gleich: **Komma**

```
?frucht <hatFarbe> "gelb" .  
?frucht <schmeckt> "bitter" .  
?frucht <schmeckt> "fruchtig" .
```

```
?frucht <hatFarbe> "gelb" ;  
      <schmeckt> "bitter" ,  
      "fruchtig" .
```

AUSGABE

- LIMIT: Ausgabe beschränken (z.B. nur 10 Records, hilfreich bei Tests)
- ORDER BY: Ausgabe sortieren

```
SELECT *  
WHERE  
{  
  x y ?a .  
  m n ?b .  
}  
LIMIT 10
```

```
SELECT *  
WHERE  
{  
  x y ?a .  
  m n ?b .  
}  
ORDER BY ?a DESC(?b)
```

BIND, FILTER

- YEAR: Funktion (Input: Datum, Output: Jahr)
- BIND: neue Variablen
- FILTER: Records auswählen

```
SELECT *  
WHERE  
{  
    BIND(YEAR(?datum) AS ?jahr)  
  
    FILTER(?jahr = 2023)  
}
```

PREFIX

Prefix: ein Teil am Anfang des Codes

```
SELECT *  
WHERE  
{  
  ?obs <https://ld.stadt-zuerich.ch/statistics/property/TIME> ?zeit ;  
      <https://ld.stadt-zuerich.ch/statistics/property/RAUM> ?raum .  
}
```

```
PREFIX property: <https://ld.stadt-zuerich.ch/statistics/property/>
```

```
SELECT *  
WHERE  
{  
  ?obs property:TIME ?zeit ;  
      property:RAUM ?raum .  
}
```

SPARQL: weitere Informationen

- SPARQL auf Wikibooks ([Link](#)), u.a. Funktionen ([Link](#))
- SPARQL auf W3C ([Link](#))
- Wikidata SPARQL-Tutorial (Deutsch, [Link](#))
- data.world SPARQL-Tutorial ([Link](#))
- Linked Data Engineering: SPARQL ([Link](#), Video-Serie)

2.4 Strukturen

Datenpunkte beschreiben

Vornamen der Wohnbevölkerung ([Link](#))

T_1					
Vornamen der wirtschaftlichen Wohnbevölkerung der Stadt Zürich, 2022 mit mindestens 10 Personen					
1. Vorname, männlich					
Quelle: Statistik Stadt Zürich, BVS					
Vorname	1996	1997	1998	1999	2000
Charles	2	5	2	3	1
Charlie	0		0	0	0
Chi	0		0	1	0
Chris	0	1	0	1	2
Christian	29	22	14	10	10
Christof	1	0	0	0	0

Datenpunkte beschreiben


Vornamen der Wohnbevölkerung ([Link](#))

T_1					
Vornamen der wirtschaftlichen Wohnbevölkerung der Stadt Zürich, 2022 mit mindestens 10 Personen					
1. Vorname, männlich					
Quelle: Statistik Stadt Zürich, BVS					
Vorname	1996	1997	1998	1999	2000
Charles	2	5	2	3	1
Charlie	0		0	0	0
Chi	0		0	1	0
Chris	0	1	0	1	2
Christian	29	22	14	10	10
Christof	1	0	0	0	0

Gemäss Bevölkerungsregister: Ende 2022 gab es in der Stadt Zürich 5 männliche Personen mit dem 1. Vornamen «Charles» und Jahrgang 1997, die zur [wirtschaftlichen Wohnbevölkerung](#) zählen.


Datenpunkte beschreiben

Gemäss **Bevölkerungsregister**: **Ende 2022** gab es in der **Stadt Zürich** **5 männliche** Personen mit dem **1. Vornamen** «**Charles**» und **Jahrgang 1997**, die zur **wirtschaftlichen Wohnbevölkerung** zählen



Strukturen

- Bei uns: pro Cube genau ein Observation Set
- Sich in Datenstrukturen bewegen: «follow your nose»



3 Praxis



Praxis

- Unterlagen: siehe www.stadt-zuerich.ch/lod
- Sie arbeiten selbstständig, mit Unterstützung der Tutor*innen



Daniel



Hansjörg



Judith




Klemens



Rolf

Wanderroute



Zusätzliches




eigene Ideen




[Federated Query:
Wikidata Bubble Chart](#)




Federated Query:
Wikidata Porträts




[Ranking](#)



[Geschlecht](#)



[Federated Query:
HAVING](#)



Mehrere Cubes:
Geburten und
Bevölkerung

4 Abschluss



Wer macht in der Schweiz Linked Data?

Linked Open Data

- Stadt Zürich (www.stadt-zuerich.ch/lod)
- Basel-Stadt ([Link](#))
- Weitere Institutionen ([Link](#))

CREATOR	CREATORNAME
1 < https://register.ld.admin.ch/opendataswiss/org/bundesamt-fur-energie-bfe >	"Bundesamt für Energie BFE"@de
2 < https://register.ld.admin.ch/opendataswiss/org/bundesamt-fur-gesundheit-bag >	"Bundesamt für Gesundheit BAG"@de
3 < https://ld.admin.ch/org/fsvs >	"Bundesamt für Lebensmittelsicherheit und Veterinärwesen"@de
4 < https://register.ld.admin.ch/opendataswiss/org/bundesamt-fur-statistik-bfs >	"Bundesamt für Statistik BFS"@de
5 < https://ld.admin.ch/office/VII.1.7 >	"Bundesamt für Umwelt"@de
6 < https://register.ld.admin.ch/opendataswiss/org/bundesamt-fur-umwelt-bafu >	"Bundesamt für Umwelt BAFU"@de
7 < https://register.ld.admin.ch/staatskalender/organization/10008758 >	"Bundesamt für Umwelt BAFU"@de
8 < https://ld.admin.ch/FCh >	"Bundeskanzlei"@de
9 < https://register.ld.admin.ch/staatskalender/organization/20030954 >	"Eidgenössische Elektrizitätskommission"@de
10 < https://register.ld.admin.ch/staatskalender/organization/10002463 >	"Eidgenössisches Amt für das Handelsregister"@de
11 < https://register.ld.admin.ch/opendataswiss/org/efv_finanzstatistik >	"Finanzstatistik EFV"@de
12 < https://register.ld.admin.ch/staatskalender/organization/10004387 >	"Schweizerische Nationalbibliothek"@de
13 < https://ld.admin.ch/office/II.1.4 >	"Schweizerisches Bundesarchiv"@de
14 < https://register.ld.admin.ch/opendataswiss/org/schweizerisches-bundesarchiv-bar >	"Schweizerisches Bundesarchiv BAR"@de
15 < https://register.ld.admin.ch/opendataswiss/org/staatsarchiv-kanton-zuerich >	"Staatsarchiv Kanton Zürich"@de
16 < https://ld.stadt-zuerich.ch/org/SSZ >	"Statistik Stadt Zürich"@de

LinkedDataGPT

- <https://ld.gpt.liip.ch>
- [Beschreibung](#) bei Liip
- Beispiel: Auswertung zu Vornamen ([Link](#))

LinkedDataGPT

Ich bin LinkedDataGPT und ich weiss (fast) alles, was auf ld.stadt-zuerich.ch publiziert ist.

Frag mich etwas über die Verwaltung der Stadt Zürich.

Besten Dank!

Zusätzliche Folien

Suchen: Cubes zu Namen?



- Daten von Statistik Stadt Zürich
- Welche Cubes zu «Namen»?
- <https://s.zazuko.com/2nongqYC>
- Vornamen der Wohnbevölkerung:
in Cube 000437

```
1 PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
2 PREFIX schema: <http://schema.org/>
3 PREFIX schemac: <https://cube.link/>
4
5 SELECT DISTINCT ?cu ?cuLabel
6 FROM <https://lindas.admin.ch/stadtzuerich/stat>
7 WHERE {
8
9     #Cubes und Cube-Labels
10    ?cu rdf:type schemac:Cube ;
11        schema:name ?cuLabel .
12
13    #Text filtern
14    FILTER REGEX(LCASE(STR(?cuLabel)), "name") .
```

	cu	cuLabel
1	https://ld.stadt-zuerich.ch/statistics/000361	Lebendgeborene nach erstem Vornamen und Geschlecht
2	https://ld.stadt-zuerich.ch/statistics/000270	Wirtschaftliche Bevölkerung nach Nachnamen
3	https://ld.stadt-zuerich.ch/statistics/000437	Wirtschaftliche Bevölkerung nach erstem Vornamen, Geburtsjahr und Geschlecht



Startpunkt: Cube zu Vornamen der Wohnbevölkerung



- Observations von Cube 000437 ausgeben lassen
- <https://s.zazuko.com/ifDzQi>

	obsSet	obs
1	https://ld.stadt-zuerich.ch/statistics/000437/observation/	https://ld.stadt-zuerich.ch/statistics/000437/observation/ Z31122021
2	https://ld.stadt-zuerich.ch/statistics/000437/observation/	https://ld.stadt-zuerich.ch/statistics/000437/observation/ Z31122021
3	https://ld.stadt-zuerich.ch/statistics/000437/observation/	https://ld.stadt-zuerich.ch/statistics/000437/observation/ Z31122021

```
1 PREFIX schema: <http://schema.org/>
2 PREFIX schemac: <https://cube.link/>
3 PREFIX datacubes: <https://ld.stadt-zuerich.ch/statistics/>
4
5 SELECT DISTINCT *
6 FROM <https://lindas.admin.ch/stadtzuerich/stat>
7 WHERE {
8
9     #Fuer den Cube 000437 (Vornamen der Wohnbevölkerung): ObservationSet
10     datacubes:000437 schemac:observationSet ?obsSet .
11
12     #Observations des ObservationSets
13     ?obsSet schemac:observation ?obs .
14
15 }
16
17 LIMIT 10
```


Observations anschauen: follow your nose



- Observation
 - Eine Observation auswählen (zum Beispiel dieser [Link](#))
 - Hinweis: in neuem Tab öffnen mit «ctrl + klick»
 - Möglichst vieles anschauen

BEW	2
GEJ	GEJ1952
NAF	NAF0001
NAM	NAM0BA0
RAUM	R30000
SEX	SEX0001
TIME	2021-12-31 (date)
ZEIT	Z31122021

Zeit: Jahr auswählen




– Zeit

- Schritt 1: Variable ?time ausgeben: <https://s.zazuko.com/2WzfZYd>

```
1 PREFIX schema: <http://schema.org/>
2 PREFIX schemac: <https://cube.link/>
3 PREFIX property: <https://ld.stadt-zuerich.ch/statistics/property/>
4 PREFIX datacubes: <https://ld.stadt-zuerich.ch/statistics/>
5
6 SELECT DISTINCT *
7 FROM <https://lindas.admin.ch/stadtzuerich/stat>
8 WHERE {
9
10     #Observation fuer ausgewaehlten Cube
11     datacubes:000437 schemac:observationSet ?obsSet .
12     ?obsSet schemac:observation ?obs .
13
14     #Zeit
15     ?obs <https://ld.stadt-zuerich.ch/statistics/property/TIME> ?time .
16
17 }
18
19 LIMIT 10
```



Zeit: Jahr auswählen




– Zeit

- Schritt 2: mit PREFIX arbeiten: <https://s.zazuko.com/2xmQqaX>

```
1 PREFIX schema: <http://schema.org/>
2 PREFIX schemac: <https://cube.link/>
3 PREFIX property: <https://ld.stadt-zuerich.ch/statistics/property/>
4 PREFIX datacubes: <https://ld.stadt-zuerich.ch/statistics/>
5
6 SELECT DISTINCT *
7 FROM <https://lindas.admin.ch/stadtzuerich/stat>
8 WHERE {
9
10     #Observation fuer ausgewaehlten Cube
11     datacubes:000437 schemac:observationSet ?obsSet .
12     ?obsSet schemac:observation ?obs .
13
14     #Zeit
15     ?obs property:TIME ?time .
16
17 }
18
19 LIMIT 10
```



Zeit: Jahr auswählen




– Zeit

- Schritt 3: aus Zeit die Variablen Jahr erstellen; mit BIND
- Schritt 4: nur das Jahr 2021 filtern: <https://s.zazuko.com/2y5pRqW>

```
1 PREFIX schema: <http://schema.org/>
2 PREFIX schemac: <https://cube.link/>
3 PREFIX property: <https://ld.stadt-zuerich.ch/statistics/property/>
4 PREFIX datacubes: <https://ld.stadt-zuerich.ch/statistics/>
5
6 SELECT DISTINCT *
7 FROM <https://lindas.admin.ch/stadtzuerich/stat>
8 WHERE {
9
10     #Observation fuer ausgewaehlten Cube
11     datacubes:000437 schemac:observationSet ?obsSet .
12     ?obsSet schemac:observation ?obs .
13
14     #Zeit
15     ?obs property:TIME ?time .
16
17     #Neue Variablen: Jahr
18     BIND(YEAR(?time) AS ?year)
19
20     #Jahr auswaehlen
21     FILTER(?year = 2021)
22 }
23
24
25 LIMIT 10
```

Raum

- Verschiedene Raum-Einheiten?
- Mit SELECT DISTINCT prüfen
- Antwort: Nein, nur für die gesamte Stadt Zürich
- <https://s.zazuko.com/34DUAUr>




```
1 PREFIX schema: <http://schema.org/>
2 PREFIX schemac: <https://cube.link/>
3 PREFIX property: <https://ld.stadt-zuerich.ch/statistics/property/>
4 PREFIX datacubes: <https://ld.stadt-zuerich.ch/statistics/>
5
6 SELECT DISTINCT ?raum
7 FROM <https://lindas.admin.ch/stadtzuerich/stat>
8 WHERE {
9
10 #Observation fuer ausgewaehlten Cube
11 datacubes:000437 schemac:observationSet ?obsSet .
12 ?obsSet schemac:observation ?obs .
13
14 #Zeit und Raum
15 ?obs property:TIME ?time ;
16 property:RAUM ?raum .
17
18 #Neue Variablen: Jahr
19 BIND(YEAR(?time) AS ?year)
20
21 #Jahr auswaehlen
22 FILTER(?year = 2021)
23
24 }
```



raum	
1	https://ld.stadt-zuerich.ch/statistics/code/R30000

Vorname



– Vorname inkl. Label

– <https://s.zazuko.com/3k7fG4u>

	nameLabel	obs
1	Charles	https://ld.stadt-zuerich.ch/statistics/00043
2	Charlotte	https://ld.stadt-zuerich.ch/statistics/00043
3	Christa	https://ld.stadt-zuerich.ch/statistics/00043
4	Nedim	https://ld.stadt-zuerich.ch/statistics/00043
5	Nelli	https://ld.stadt-zuerich.ch/statistics/00043



```
1 PREFIX schema: <http://schema.org/>
2 PREFIX schemac: <https://cube.link/>
3 PREFIX property: <https://ld.stadt-zuerich.ch/statistics/property/>
4 PREFIX datacubes: <https://ld.stadt-zuerich.ch/statistics/>
5
6 SELECT ?nameLabel ?obs
7 FROM <https://lindas.admin.ch/stadtzuerich/stat>
8 WHERE {
9
10 #Observation fuer ausgewaehlten Cube
11   datacubes:000437 schemac:observationSet ?obsSet .
12   ?obsSet schemac:observation ?obs .
13
14 #Properties
15   ?obs property:TIME ?time ;
16         property:NAM ?name .
17
18 #Labels
19   ?name schema:name ?nameLabel .
20
21 #Neue Variablen: Jahr
22   BIND(YEAR(?time) AS ?year)
23
24 #Jahr auswaehlen
25   FILTER(?year = 2021)
26
27 }
28
29 LIMIT 10
```

Vorsicht

- Anzahl Personen pro Name?
- Ist die Auswertung korrekt?
 - Nein, das Geburtsjahr fehlt
 - <https://s.zazuko.com/3BuYmte>



nameLabel	pers
Charles	"2.0"^^xsd:double
Charlotte	"4.0"^^xsd:double
Christa	"8.0"^^xsd:double
Nedim	"1.0"^^xsd:double
Nelli	"1.0"^^xsd:double
Nicholas	"1.0"^^xsd:double
Nicolas	"1.0"^^xsd:double
Olga	"1.0"^^xsd:double

```
1 PREFIX schema: <http://schema.org/>
2 PREFIX schemac: <https://cube.link/>
3 PREFIX property: <https://ld.stadt-zuerich.ch/statistics/property/>
4 PREFIX measure: <https://ld.stadt-zuerich.ch/statistics/measure/>
5 PREFIX datacubes: <https://ld.stadt-zuerich.ch/statistics/>
6
7 SELECT ?nameLabel ?pers ?obs
8 FROM <https://lindas.admin.ch/stadtzuerich/stat>
9 WHERE {
10
11 #Observation fuer ausgewaehlten Cube
12   datacubes:000437 schemac:observationSet ?obsSet .
13   ?obsSet schemac:observation ?obs .
14
15 #Properties
16   ?obs property:TIME ?time ;
17         property:NAM ?name .
18
19 #Labels
20   ?name schema:name ?nameLabel .
21
22 #Measure: Personen
23   ?obs measure:BEW ?pers .
24
25 #Neue Variablen: Jahr
26   BIND(YEAR(?time) AS ?year)
27
28 #Jahr auswaehlen
29   FILTER(?year = 2021)
30
31 }
32
33 LIMIT 10
```

Geburtsjahr

– <https://s.zazuko.com/3gyjgAH>


nameLabel	gebjahr	pers
Charlotte	1952	"4.0"^^xsd:double
Charlotte	2018	"11.0"^^xsd:double
Charlotte	1935	"2.0"^^xsd:double
Charlotte	1942	"10.0"^^xsd:double
Charlotte	1954	"4.0"^^xsd:double
Charlotte	1934	"2.0"^^xsd:double
Charlotte	1938	"4.0"^^xsd:double

```
1 PREFIX schema: <http://schema.org/>
2 PREFIX schemac: <https://cube.link/>
3 PREFIX property: <https://ld.stadt-zuerich.ch/statistics/property/>
4 PREFIX measure: <https://ld.stadt-zuerich.ch/statistics/measure/>
5 PREFIX datacubes: <https://ld.stadt-zuerich.ch/statistics/>
6
7 SELECT DISTINCT ?nameLabel ?gebjahr ?pers ?obs
8 FROM <https://lindas.admin.ch/stadtzuerich/stat>
9 WHERE {
10
11     #Observation fuer ausgewaehlten Cube
12     datacubes:000437 schemac:observationSet ?obsSet .
13     ?obsSet schemac:observation ?obs .
14
15     #Properties
16     ?obs property:TIME ?time ;
17         property:NAM ?name ;
18         property:GEJ ?geburtsjahr .
19
20     #Geburtsjahr
21     ?geburtsjahr schema:position ?gebjahr .
22
23     #Name: Label
24     ?name schema:name ?nameLabel .
25
26     #Personen
27     ?obs measure:BEW ?pers .
28
29     #Jahr
30     BIND(YEAR(?time) AS ?year)
31
32     #Auswahl
33     FILTER((?year = 2021) && (?nameLabel = 'Charlotte'))
34
35 }
36
37 LIMIT 10
```




Grafik


- Anzahl pro Geburtsjahr
- Beispiel: Michael
- <https://s.zazuko.com/23svbCk>



```
1 PREFIX schema: <http://schema.org/>
2 PREFIX schemac: <https://cube.link/>
3 PREFIX property: <https://ld.stadt-zuerich.ch/statistics/property/>
4 PREFIX measure: <https://ld.stadt-zuerich.ch/statistics/measure/>
5 PREFIX datacubes: <https://ld.stadt-zuerich.ch/statistics/>
6
7 SELECT DISTINCT ?nameLabel ?gebjahr ?pers ?obs
8 FROM <https://lindas.admin.ch/stadtzuerich/stat>
9 WHERE {
10
11   #Observation fuer ausgewaehlten Cube
12   datacubes:000437 schemac:observationSet ?obsSet .
13   ?obsSet schemac:observation ?obs .
14
15   #Properties
16   ?obs property:TIME ?time ;
17   property:NAM ?name ;
18   property:GEJ ?geburtsjahr .
19
20   #Geburtsjahr
21   ?geburtsjahr schema:position ?gebjahr .
22
23   #Name: Label
24   ?name schema:name ?nameLabel .
25
26   #Personen
27   ?obs measure:BEW ?pers .
28
29   #Jahr
30   BIND(YEAR(?time) AS ?year)
31
32   #Auswahl
33   FILTER((?year = 2021) && (?nameLabel = 'Michael'))
34
35 }
36
37 ORDER BY DESC(?gebjahr)
```




Supplement: mit Geschlecht



– Beispiel: Andrea

– <https://s.zazuko.com/3rqGCX9>



```
1 PREFIX schema: <http://schema.org/>
2 PREFIX schemac: <https://cube.link/>
3 PREFIX property: <https://ld.stadt-zuerich.ch/statistics/property/>
4 PREFIX measure: <https://ld.stadt-zuerich.ch/statistics/measure/>
5 PREFIX datacubes: <https://ld.stadt-zuerich.ch/statistics/>
6
7 SELECT DISTINCT ?nameLabel ?gebjahr ?geslLabel ?pers ?obs
8 FROM <https://lindas.admin.ch/stadtzuerich/stat>
9 WHERE {
10
11 #Observation fuer ausgewaehlten Cube
12 datacubes:000437 schemac:observationSet ?obsSet .
13 ?obsSet schemac:observation ?obs .
14
15 #Properties
16 ?obs property:TIME ?time ;
17 property:NAM ?name ;
18 property:SEX ?gesl ;
19 property:GEJ ?geburtsjahr .
20
21 #Geburtsjahr
22 ?geburtsjahr schema:position ?gebjahr .
23
24 #Name: Labels
25 ?name schema:name ?nameLabel .
26 ?gesl schema:name ?geslLabel .
27
28 #Personen
29 ?obs measure:BEW ?pers .
30
31 #Jahr
32 BIND(YEAR(?time) AS ?year)
33
34 #Auswahl
35 FILTER((?year = 2021) && (?nameLabel = 'Andrea'))
36
37 }
38
39 ORDER BY DESC(?gebjahr)
```

Supplement: Ranking

- häufigste Frauennamen
- [Link](#)

	nameLabel	total
1	Maria	"3946.0"^^xsd:double
2	Anna	"2694.0"^^xsd:double
3	Laura	"1483.0"^^xsd:double
4	Claudia	"1476.0"^^xsd:double
5	Andrea	"1453.0"^^xsd:double
6	Barbara	"1417.0"^^xsd:double
7	Ursula	"1382.0"^^xsd:double
8	Sandra	"1317.0"^^xsd:double
9	Ruth	"1310.0"^^xsd:double
10	Monika	"1246.0"^^xsd:double


```
1 PREFIX schema: <http://schema.org/>
2 PREFIX schemac: <https://cube.link/>
3 PREFIX property: <https://ld.stadt-zuerich.ch/statistics/property/>
4 PREFIX measure: <https://ld.stadt-zuerich.ch/statistics/measure/>
5 PREFIX datacubes: <https://ld.stadt-zuerich.ch/statistics/>
6
7 SELECT ?nameLabel (SUM(?pers) AS ?total)
8 FROM <https://lindas.admin.ch/stadtzuerich/stat>
9 WHERE {
10
11 #Observation fuer ausgewaehlten Cube
12 datacubes:000437 schemac:observationSet ?obsSet .
13 ?obsSet schemac:observation ?obs .
14
15 #Properties
16 ?obs property:TIME ?time ;
17 property:NAM ?name ;
18 property:SEX ?gesl .
19
20 #Name: Labels
21 ?name schema:name ?nameLabel .
22 ?gesl schema:name ?geslLabel .
23
24 #Personen
25 ?obs measure:BEW ?pers .
26
27 #Jahr
28 BIND(YEAR(?time) AS ?year)
29
30 #Auswahl
31 FILTER((?year = 2021) && (?geslLabel = 'weiblich'))
32
33 }
34
35 GROUP BY(?nameLabel)
36 ORDER BY DESC(?total) ?nameLabel
```



Supplement: Federated Query, Wikidata Bubble Chart




- Wikidata
 - Von Wikidata auf unsere Daten zugreifen
 - Tipp: SERVICE
 - <https://w.wiki/7EeP>
 - Es dauert etwa 10 Sekunden



Supplement: Federated Query, HAVING



- Nur Namen, die mindestens 1000 Mal vorkommen
- Tipp: HAVING
- <https://w.wiki/7EeS>




```
1 PREFIX schema: <http://schema.org/>
2 PREFIX schemac: <https://cube.link/>
3 PREFIX property: <https://ld.stadt-zuerich.ch/statistics/property/>
4 PREFIX measure: <https://ld.stadt-zuerich.ch/statistics/measure/>
5 PREFIX datacubes: <https://ld.stadt-zuerich.ch/statistics/>
6
7 SELECT ?nameLabel (SUM(?pers) AS ?total)
8
9 WHERE {
10
11   SERVICE <https://ld.stadt-zuerich.ch/query> {
12
13     #Observation fuer ausgewaehlten Cube
14     datacubes:000437 schemac:observationSet ?obsSet .
15     ?obsSet schemac:observation ?obs .
16
17     #Properties
18     ?obs property:TIME ?time ;
19         property:NAM ?name ;
20         property:SEX ?gesl .
21
22     #Name: Labels
23     ?name schema:name ?nameLabel .
24     ?gesl schema:name ?geslLabel .
25
26     #Personen
27     ?obs measure:BEW ?pers .
28
29     #Jahr
30     BIND(YEAR(?time) AS ?year)
31
32     #Auswahl
33     FILTER((?year = 2021) && (?geslLabel = 'weiblich'))
34
35     #SERVICE Ende
36   }
37
38 }
39
40
41 GROUP BY(?nameLabel)
42 HAVING (?total >= 1000)
43 ORDER BY DESC(?total) ?nameLabel
```

Supplement: Federated Query, Wikidata Bilder



- Frauennamen: Top 10
- Bilder von Personen auf Wikidata mit Geburtsort Zürich




```

PREFIX schema: <http://schema.org/>
PREFIX schemac: <https://cube.link/>
PREFIX property: <https://ld.stadt-zuerich.ch/statistics/property/>
PREFIX measure: <https://ld.stadt-zuerich.ch/statistics/measure/>
PREFIX datacubes: <https://ld.stadt-zuerich.ch/statistics/>
PREFIX code: <https://ld.stadt-zuerich.ch/statistics/code/>
7
8 #Wikidata
9
10 SELECT ?nameLabel ?total ?person ?personLabel ?pic
11
12 WHERE {
13
14   ?person wdt:P21 wd:Q6581072 ;
15           wdt:P19 wd:Q72 ;
16           wdt:P18 ?pic ;
17           wdt:P735 ?name .
18   ?name rdfs:label ?nameLabel .
19
20   FILTER((LANG(?nameLabel)) = "de")
21   FILTER((STR(?nameLabel)) = ?nameLabelSSZ)
22
23
24 #Stadt Zuerich
25
26 {
27
28   SELECT ?nameLabelSSZ (SUM(xsd:integer(?pers)) AS ?total)
29
30   WHERE {
31
32     SERVICE <https://ld.stadt-zuerich.ch/query> {
33
34       datacubes:000437 schemac:observationSet ?obsSet .
35       ?obsSet schemac:observation ?obs .
36       ?obs property:TIME ?time ;
37            property:NAM ?name ;
38            property:SEX code:SEX0002 ;
39            measure:BEW ?pers .
40       ?name schema:name ?nameLabelSSZ .
41
42       BIND(YEAR(?time) AS ?year)
43       FILTER(?year = 2021)
44
45     }
46   }
47   GROUP BY(?nameLabelSSZ)
48   ORDER BY DESC(?total)
49   LIMIT 10
50
51 }
52
53 SERVICE wikibase:label { bd:serviceParam wikibase:language "de, en, fr, it" . }
54
55 }
56
57 ORDER BY DESC(?total)
    
```

```


1 PREFIX schema: <http://schema.org/>
2 PREFIX schemac: <https://cube.link/>
3 PREFIX property: <https://ld.stadt-zuerich.ch/statistics/property/>
4 PREFIX measure: <https://ld.stadt-zuerich.ch/statistics/measure/>
5 PREFIX datacubes: <https://ld.stadt-zuerich.ch/statistics/>
6 PREFIX code: <https://ld.stadt-zuerich.ch/statistics/code/>
7
8 #Wikidata
9
10 SELECT ?nameLabel ?total ?person ?personLabel ?pic
11
12 WHERE {
13
14   ?person wdt:P21 wd:Q6581072 ;
15           wdt:P19 wd:Q72 ;
16           wdt:P18 ?pic ;
17           wdt:P735 ?name .
18   ?name rdfs:label ?nameLabel .
19
20   FILTER((LANG(?nameLabel)) = "de")
21   FILTER((STR(?nameLabel)) = ?nameLabelSSZ)
22
23
24 #Stadt Zuerich
25
26 {
27
28   SELECT ?nameLabelSSZ (SUM(xsd:integer(?pers)) AS ?total)
29
30   WHERE {
31
32     SERVICE <https://ld.stadt-zuerich.ch/query> {
33
34       datacubes:000437 schemac:observationSet ?obsSet .
35       ?obsSet schemac:observation ?obs .
36       ?obs property:TIME ?time ;
37            property:NAM ?name ;
38            property:SEX code:SEX0002 ;
39            measure:BEW ?pers .
40       ?name schema:name ?nameLabelSSZ .
41
42       BIND(YEAR(?time) AS ?year)
43       FILTER(?year = 2021)
44
45     }
46   }
47   GROUP BY(?nameLabelSSZ)
48   ORDER BY DESC(?total)
49   LIMIT 10
50
51 }
52
53 SERVICE wikibase:label { bd:serviceParam wikibase:language "de, en, fr, it" . }
54
55 }
56
57 ORDER BY DESC(?total)
    
```

Supplement: Mehrere Cubes, Geburten und Bevölkerung



- Mehrere Cubes: männliche Vornamen
 - Geburten im Jahr 2021
 - Bevölkerung per Ende 2021
 - Welches sind «neue» Vornamen?
- Beispiel Theo
 - 124 Theos gab es Ende 2021 in Zürich
 - 20 Theos wurden 2021 geboren
 - 16 Prozent im Jahr 2021 geboren
- Beispiel Alexander
 - 1399 Alexander gab es Ende 2021 in Zürich
 - 14 Alexander wurden 2021 geboren
 - 1 Prozent im Jahr 2021 geboren

Supplement: Mehrere Cubes, Geburten und Bevölkerung



	nameLabel	geburten	personen	prozent
1	Noah	"27"	"413"	"7"
2	Leo	"24"	"387"	"6"
3	Louis	"22"	"369"	"6"
4	Theo	"20"	"124"	"16"
5	Levi	"18"	"106"	"17"
6	Luca	"18"	"960"	"2"
7	Mateo	"18"	"118"	"15"
8	Luis	"17"	"453"	"4"
9	Maximilian	"17"	"501"	"3"
10	Felix	"16"	"827"	"2"
11	Finn	"16"	"483"	"9"
12	Leonardo	"16"	"284"	"6"
13	Liam	"16"	"202"	"8"
14	Daniel	"15"	"3052"	"0"
15	Elias	"15"	"324"	"5"
16	Jan	"15"	"981"	"2"
17	Vincent	"15"	"357"	"4"
18	Alexander	"14"	"1399"	"1"
19	Gabriel	"14"	"500"	"3"
20	Leon	"14"	"331"	"4"

```
# in SPARQL-Editor der Stadt Zuerich

PREFIX schema: <http://schema.org/>
PREFIX schemac: <https://cube.link/>
PREFIX property: <https://ld.stadt-zuerich.ch/statistics/property/>
PREFIX measure: <https://ld.stadt-zuerich.ch/statistics/measure/>
PREFIX datacubes: <https://ld.stadt-zuerich.ch/statistics/>
PREFIX code: <https://ld.stadt-zuerich.ch/statistics/code/>

SELECT ?nameLabel ?geburten ?personen (ROUND(?geburten/?personen * 100) AS ?prozent)
FROM <https://lindas.admin.ch/stadtzuerich/stat>
WHERE {
  {
    SELECT ?name (SUM(xsd:integer(?geb)) AS ?geburten)
    WHERE {
      #Geburten (G): Observations
      datacubes:000361 schemac:observationSet ?obsSetG .
      ?obsSetG schemac:observation ?obsG .

      #Properties
      ?obsG property:TIME "2021-12-31"^^xsd:date ;
        property:SEX code:SEX0001 ;
        property:NAF code:NAF0001 ;
        property:NAM ?name .

      #Geburten
      ?obsG measure:GEB ?geb .
    }
  }
  GROUP BY ?name
}
OPTIONAL
{
  SELECT ?name (SUM(xsd:integer(?pers)) AS ?personen)
  WHERE {
    #Bevoelkerung (B): Observations
    datacubes:000437 schemac:observationSet ?obsSetB .
    ?obsSetB schemac:observation ?obsB .

    #Properties
    ?obsB property:TIME "2021-12-31"^^xsd:date ;
      property:SEX code:SEX0001 ;
      property:NAM ?name .

    #Personen
    ?obsB measure:BEW ?pers .
  }
}
GROUP BY ?name

#Labels
?name schema:name ?nameLabel .

ORDER BY DESC(?geburten) ?nameLabel
LIMIT 20
```

```
1 PREFIX schema: <http://schema.org/>
2 PREFIX schemac: <https://cube.link/>
3 PREFIX property: <https://ld.stadt-zuerich.ch/statistics/property/>
4 PREFIX measure: <https://ld.stadt-zuerich.ch/statistics/measure/>
5 PREFIX datacubes: <https://ld.stadt-zuerich.ch/statistics/>
6 PREFIX code: <https://ld.stadt-zuerich.ch/statistics/code/>
7
8 SELECT ?nameLabel ?geburten ?personen
9 FROM <https://lindas.admin.ch/stadtzuerich/stat>
10 WHERE {
11 {
12 SELECT ?name (SUM(?geb) AS ?geburten)
13 WHERE {
14
15 #Geburten (G): Observations
16 datacubes:000361 schemac:observationSet ?obsSetG .
17 ?obsSetG schemac:observation ?obsG .
18
19 #Properties
20 ?obsG property:TIME "2021-12-31"^^xsd:date ;
21 property:SEX code:SEX0001 ;
22 property:NAF code:NAF0001 ;
23 property:NAM ?name .
24
25 #Geburten
26 ?obsG measure:GEB ?geb .
27
28 }
29 GROUP BY ?name
30 }
31 OPTIONAL
32 {
33 SELECT ?name (SUM(?pers) AS ?personen)
34 WHERE {
35
36 #Bevoelkerung (B): Observations
37 datacubes:000437 schemac:observationSet ?obsSetB .
38 ?obsSetB schemac:observation ?obsB .
39
40 #Properties
41 ?obsB property:TIME "2021-12-31"^^xsd:date ;
42 property:SEX code:SEX0001 ;
43 property:NAM ?name .
44
45 #Personen
46 ?obsB measure:BEW ?pers .
47
48 }
49 GROUP BY ?name
50
51 }
52
53 #Labels
54 ?name schema:name ?nameLabel .
55
56 }
57
58 ORDER BY DESC(?geburten)
```